

## 加速度記録の基線補正 CRAC

プログラムCRAC (Base-line Correction of Accelerogram) は、与えられた加速度時刻歴に対し、基線を調整することによって、継続時間終了時の速度が0となり、残留変位が不当に過大な値とならないよう、これを補正するサブルーチン副プログラムである。

### CRAC (加速度記録の補正)

#### 【目的】

基線を補正して、継続時間終了時に速度が0となり、残留変位が相応な値に収まるよう、加速度時刻歴を修正する。

#### 【使用法】

##### (1) 接続方法

CALL CRAC (DT, NN, DDYMAX, DDY, ND, VW1, VW2)

引数	型	プログラムを呼ぶときの内容	プログラムから戻ったときの内容
DT	R	時刻歴の時間間隔 (単位 sec)	不変
NN	I	時刻歴データの個数	不変
DDYMAX	R	最大加速度 (単位 gal)	不変
DDY	R 1次元配列 (ND)	与えられた加速度時刻歴 (単位 gal)	補正された加速度時刻歴 (単位 gal)
ND	I	配列 DDY, VW1, VW2 の整合寸法	不変
VW1	R 1次元配列 (ND)	何も入れなくてよい	(作業領域)
VW2	R 1次元配列 (ND)	何も入れなくてよい	(作業領域)

##### (2) 必要なサブルーチンおよび関数副プログラム

IACC

## 【計算法】

まず与えられた加速度時刻歴  $\ddot{y}(t)$  を，サブルーチン IACC を呼んで積分し，速度と変位の時刻歴  $\dot{y}(t)$ ， $y(t)$  を求める．そこで変位，速度，加速度の修正値  $\hat{y}(t)$ ， $\hat{\dot{y}}(t)$ ， $\hat{\ddot{y}}(t)$  を

$$\left. \begin{aligned} \hat{y}(t) &= y(t) - \left( \frac{1}{2} a_0 t^2 + \frac{1}{6} a_1 t^3 \right) \\ \hat{\dot{y}}(t) &= \dot{y}(t) - \left( a_0 t + \frac{1}{2} a_1 t^2 \right) \\ \hat{\ddot{y}}(t) &= \ddot{y}(t) - (a_0 + a_1 t) \end{aligned} \right\} \quad (a)$$

によって表わすものとし，継続時間を  $T$  とすれば，(a) 式の第 2 式において  $\dot{y}(T) = 0$  となる条件は

$$a_0 = \frac{\dot{y}(T)}{T} - \frac{a_1 T}{2} \quad (b)$$

であり，これより

$$\frac{da_0}{da_1} = -\frac{T}{2} \quad (c)$$

が得られる．

$t = T$  における残留変位  $\hat{y}(T)$  が，不当に過大とまらない条件を満たすためには，(a) 式の第 1 式で右辺の ( ) 内にある多項式が，曲線  $y(t)$  に最もよく適合するよう，最小 2 乗法により

$$\varepsilon = \int_0^T \left[ y(t) - \left( \frac{1}{2} a_0 t^2 + \frac{1}{6} a_1 t^3 \right) \right]^2 dt$$

とにおいて

$$\frac{d\varepsilon}{da_1} = 0 \quad (d)$$

とする．

(b)，(c)，(d) 式から係数  $a_1$  は

$$a_1 = \frac{28}{13} \cdot \frac{1}{T^2} \left[ 2\dot{y}(T) - \frac{15}{T^5} \int_0^T y(T)(3Tt^2 - 2t^3) dt \right] \quad (e)$$

と表わされる．このプログラムでは，(e) 式右辺の積分を実行するのに，特に高い精度を要求される計算でもないから，最も簡単な台形法則によっている．

(e) 式によって  $a_1$  が求まれば， $a_0$  も (b) 式によって定まるので，補正加速度時刻歴  $\hat{\ddot{y}}(t)$  を，(a) 式の第 3 式によって計算する．

ただし，補正を行った加速度時刻歴の最大値  $|\hat{\ddot{y}}(t)|_{\max}$  は，もとの時刻歴の最大値  $|\ddot{y}(t)|_{\max}$  とは，若干異なった値となっているので，補正した時刻歴  $\hat{\ddot{y}}(t)$  の全体に，係数

$C = |\ddot{y}(t)|_{\max} / |\hat{\ddot{y}}(t)|_{\max}$  を掛けて，最大加速度をもとの値に戻す．

## 【プログラム】

C	* * * * *	CRAC	1
C	SUBROUTINE FOR BASE-LINE CORRECTION OF ACCELEROGRAM	CRAC	2
C	* * * * *	CRAC	3
C		CRAC	4
C	CODED BY Y.OHSAKI	CRAC	5
C		CRAC	6
C	SUBROUTINE CRAC(DT, NN, DDYMAX, DDY, ND, VW1, VW2)	CRAC	7
C		CRAC	8
C	DIMENSION DDY(ND), VW1(ND), VW2(ND)	CRAC	9
C		CRAC	10
	CALL IACC(DT, NN, DDY, VW1, VW2, ND, DUMMY, DUMMY)	CRAC	11
	TT=REAL(NN-1)*DT	CRAC	12
	T=0.	CRAC	13
	DO 110 M=1, NN	CRAC	14
	VW2(M)=VW2(M)*(3.*TT-2.*T)*T**2	CRAC	15
	T=T+DT	CRAC	16
110	CONTINUE	CRAC	17
	SUM=(VW2(1)+VW2(NN))/2.	CRAC	18
	DO 120 M=2, NN-1	CRAC	19
	SUM=SUM+VW2(M)	CRAC	20
120	CONTINUE	CRAC	21
	SUM=SUM*DT	CRAC	22
	A1=28./13./TT**2*(2.*VW1(NN)-15./TT**5*SUM)	CRAC	23
	A0=VW1(NN)/TT-A1/2.*TT	CRAC	24
	T=0.	CRAC	25
	ACMAX=0.	CRAC	26
	DO 130 M=1, NN	CRAC	27
	DDY(M)=DDY(M)-A0-A1*T	CRAC	28
	ACMAX=AMAX1(ACMAX, ABS(DDY(M)))	CRAC	29
	T=T+DT	CRAC	30
130	CONTINUE	CRAC	31
	COEF=DDYMAX/ACMAX	CRAC	32
	DO 140 M=1, NN	CRAC	33
	DDY(M)=DDY(M)*COEF	CRAC	34
140	CONTINUE	CRAC	35
	RETURN	CRAC	36
	END	CRAC	37